
1. ОСНОВНЫЕ ПОНЯТИЯ АЛГОРИТМИЗАЦИИ И АЛГОРИТМИЧЕСКОГО ЯЗЫКА QBASIC
1.1. Этапы подготовки задачи к решению на компьютере

Процесс подготовки любой задачи к ее решению на компьютере состоит из ряда последовательных этапов:

· постановка задачи;

· алгоритмизация;

· программирование;

· отладка программы.

Постановка задачи представляет собой процесс построения ее математи​ческой модели. На этом этапе формулируются цель и условия решения задачи, подробно описывается ее содержание.

Алгоритмизация —- это процесс построения алгоритма задачи.

Под программированием понимается процесс кодирования составленного алгоритма на одном из языков программирования.

Отладка программы предполагает исправление синтаксических и семантических (смысловых) ошибок в тексте программы и проверку работоспособности программы на контрольном примере.

Считая постановку рассматриваемых в данном пособии задач уже заданной, подробно остановимся на вопросах алгоритмизации и программирования.

1.2. Алгоритмы, их свойства и изображение

Алгоритмом называется точное и понятное предписание исполнителю совершить последовательность действий, направленных на решение поставленной задачи.

Слово «алгоритм» происходит от имени математика IX века 
Аль-Хорезми, который сформулировал правила выполнения арифметических действий.

Первоначально под алгоритмом и понимали только правила выполнения четырех арифметических действий над числами. В дальнейшем это понятие стали использовать вообще для обозначения последовательности действий, приводящих к решению любой поставленной задачи.

Говоря об алгоритме вычислительного процесса, необходимо понимать, что объектами, к которым применяется алгоритм, являются данные. Алгоритм решения вычислительной задачи представляет собой совокупность правил преобразования исходных данных в результатные (рис.1).

Рис.1. Представление алгоритма вычислительного процесса

Основными свойствами алгоритма являются:

1. Детерминированность (определенность). Предполагает получение одно​значного результата вычислительно процесса при заданных исходных данных. Благодаря этому свойству процесс выполнения алгоритма носит механический ха​рактер.

2. Результативность. Указывает на наличие таких исходных данных, для которых реализуемый по заданному алгоритму вычислительный процесс должен через конечное число шагов остановиться и выдать искомый результат.

3. Массовость. Это свойство предполагает, что алгоритм должен быть пригоден для решения всех задач данного типа.

4. Дискретность. Означает расчлененность определяемого алгоритмом вычислительного процесса на отдельные этапы, возможность выполнения которых исполнителем (компьютером) не вызывает сомнений.

Алгоритм должен быть формализован по некоторым правилам посредством конкретных изобразительных средств. К ним относятся следующие способы записи алгоритмов: словесный, формульно-словесный, графический, язык операторных схем, алгоритмический язык.

Наибольшее распространение благодаря своей наглядности получил графиче​ский (блок-схемный) способ записи алгоритмов.

Блок-схемой называется графическое изображение логической структуры алго​ритма, в котором каждый этап процесса обработки информации представляется в виде геометрических символов (блоков), имеющих определенную конфигурацию в зависимости от характера выполняемых операций.

Перечень символов, их наименование, отображаемые ими функции, форма и размеры определяются ГОСТ 19.003-80, ГОСТ 19.002-80 и ГОСТ 19701-90.

Не вдаваясь во все тонкости проектирования блок-схем алгоритмов, используем обозначения некоторых символов для описания основных этапов рассматриваемых вычислительных процессов (табл.1).

                                                                                                  Таблица 1

Некоторые символы блок-схем

	Символ
	Наименование символа
	Функция

	

	Ввод-вывод
	Преобразование данных в форму, пригодную для обработки (ввод) или отображение результатов обработки (вывод)

	

	Процесс


	Выполнение операций, в результате которых изменяется значение, форма представления или расположение данных

	

	Решение
	Выбор направления выполнения алгоритма в зависимости от некоторых переменных условий

	
	Пуск-останов
	Начало, конец, прерывание процесса обработки

	

	Предопределенный процесс
	Использование ранее созданных и отдельно записанных алгоритмов


1.3. Основные понятия QBASIC

Название языка программирования BASIC — это аббревиатура английских слов Begginer’s All-purpose Symbolic Instruction Code (всецелевой язык программиро​вания для начинающих).

BASIC (БЕЙСИК) был разработан в 1963 году профессорами  Дартмундского колледжа Джоном Кемени и Томасом Куртцом. Сегодня BASIC превратился в современный язык высокого уровня, хотя продолжает оставаться простым и доступным для всех пользователей.

Самое широкое распространение получили версии BASIC, разработанные фирмой Microsoft. Существуют три поколения языка, разработанные этой фирмой:

· поколение GWBASIC, BASICа;

· поколение QuickBASIC;

· поколение Visual BASIC.

Начиная с версии MS DOS 5.0, фирма Microsoft поставляет интерпретатор QBASIC, который представляет собой усеченный вариант языка QuickBASIC. В языке QBASIC отсутствует возможность создания загрузочных модулей и использование внешних программных модулей.

Алфавит языка QBASIC

Алфавит языка представляет собой набор символов, используемых для построения слов в предложениях языка. В алфавит языка QBASIC входят:

· все буквы латинского алфавита (A – Z, a – z)

· цифры (0 – 9)

· знаки операций: +, - , * , / , \ , ^ , = , < , >

· символы-разделители:     ,  .  :  ;  “  ‘  (  )  _

· символы объявления типа данных : % , & , ! , # , $

Структура программы

Основной структурной единицей программы на QBASIC является строка. Ее максимальная длина составляет 255 символов. Строки в QBASIC не нужно нумеровать. Строка программы может содержать операторы языка и коммента​рий. 

Операторы языка – это такие языковые конструкции (предложения), которые реализуют алгоритм решения задачи.

Комментарий – эта любая последовательность символов, поясняющая текст программы.

Если строка программы содержит несколько операторов, то эти операторы должны быть разделены между собой двоеточием (:).

Если строка программы содержит не только операторы, но и комментарий, то комментарий должен  размещаться после всех операторов и отделяться от них апострофом (‘).

В случае, если строка программы содержит только комментарий, то начинаться она должна или с апострофа (‘), или с ключевого слова REM.
Данные 

Данные, обрабатываемые в языке QBASIC, подразделяются на два вида:

· константы
· переменные.
Константами  называются данные, значения которых в процессе работы программы не изменяются. В табл.2 приведены примеры записи констант.

                                                                           Таблица 2
Примеры записи констант

	Обычная запись
	Запись на QBASIC

	7
	7

	1,5
	1.5

	38,2·107
	38.2 E +7

	1,3·10-8
	1.3 E -8

	Результат
	“Результат”


Для обозначения константы в тексте программы достаточно поместить ее значение.

Переменная – это величина, которая может меняться при выполнении программы. Для обозначения переменных в программе используются их имена или идентификаторы.

Идентификатор – это последовательность, содержащая до 40 символов латинских букв и цифр, начинающаяся с буквы.

В табл.3 приведены примеры записи идентификаторов.

                                                                                 Таблица 3
Примеры записи идентификаторов

	Правильная запись
	Неправильная запись

	A1
	1A (начинается с цифры)

	KODDETALI
	KOD  DETALI (использован пробел)

	M
	– M (использован символ –


Арифметические выражения

Арифметическое выражение - это совокупность числовых констант, переменных, функций, соединенных знаками арифметических операций. 

Перечислим арифметические операции в порядке убывания приоритета:

^     —     возведение в степень;

* , / — умножение, деление;

+ , –  — сложение, вычитание.

В выражении сначала выполняются операции более высокого приоритета, затем операции одного уровня слева направо. Для изменения порядка вычисления арифметического выражения используются круглые скобки.

Примеры записи арифметических выражений приведены в табл.4.

                                                                                      Таблица 4

Примеры записи арифметических выражений

	Алгебраическое выражение
	Выражение на QBASIC

	              



	       (a-b)^2/(a+b)

	                 



	            a∗b/a^3

	         


	             a∗ (–b)  

              a∗–b


В состав арифметического выражения могут входить функции. В QBASIC имеется возможность использовать предопределенные или встроенные (стандартные) функции и определять свои пользовательские функции.

Рассмотрим встроенные математические функции языка QBASIC.

ABS ( x)  –  возвращает абсолютное значение числа x.

EXP ( x) – вычисляет основание натурального логарифма (число e 
в степени x).

LOG ( x) – возвращает натуральный логарифм x  ( x>0).

SGN (x) – возвращает знак числа x. Если x>0, возвращается значение +1; если x=0, возвращается значение 0; если x<0, то возвращает значение –1. 

SQR (x) – возвращает квадратный корень числового выражения (x>=0).

ATN (x) – возвращает арктангенс числа x.

COS (x) – возвращает значение косинуса от угла в радианах  x.

SIN (x) - возвращает значение синуса от заданного угла 
в радианах x.

TAN (x) – возвращает тангенс угла в радианах . x – угол в радианах.

RND [(n)] – возвращает случайное число обычной точности в интервале от 0 до 1.

Если аргумент опущен, возвращается следующее число из последовательности случайных чисел; n определяет, как генерируется случайное число;

n<0  возвращает то же число для любого n;

n>0  возвращает следующее число из последовательности;

n=0  возвращает последнее выданное число.

Примечание. Если генератор случайных чисел не запущен, функция RND будет вызывать одну и ту же последовательность чисел при запуске программы. Для изменения последовательности надо поставить оператор RANDOMIZE  с функцией TIMER в начало программы и изменять аргумент перед каждым запуском программы  RANDOMIZE  TIMER.

Пример использования встроенных математических функций в выражении:

y = ABS (x) + SQR (z) - LOG (n)

Примечание. Функции обработки числовых  и символьных данных будут рассмотрены нами в разделе 5.
2. АЛГОРИТМИЗАЦИЯ И ПРОГРАММИРОВАНИЕ НА QBASIC ОСНОВНЫХ ВИДОВ ВЫЧИСЛИТЕЛЬНЫХ 
ПРОЦЕССОВ

При всем многообразии алгоритмов решения задач в них можно выделить три  основных вида вычислительных процессов:

· линейный;

· ветвящийся;

· циклический.

2.1. Линейные вычислительные процессы

Линейным называется такой вычислительный процесс, при котором все этапы решения задачи выполняются в естественном порядке следования записи этих этапов.

Примером линейной алгоритмической структуры может служить алгоритм решения задачи 1 со следующим условием: найти и вывести результат вычисления выражения:   

 .

На рис.2 представлена блок-схема решения этой задачи.

Так как данная схема – первая рассматриваемая в данном пособии, то объясним подробно назначение каждого из используемых в ней блоков.

Блоки 1 и 5 служат соответственно для обозначения начала и окончания вычислительного процесса. 

Основной принцип программирования заключается в том, что обрабатывать можно только те данные, которые находятся в определенных областях оперативной памяти компьютера. При задании переменным различных имен (идентификаторов) предполагается, что данные будут располагаться в различных областях оперативной памяти. 

	Для того, чтобы можно было получить результат, который по условию задачи 1 должен располагаться в области памяти Y, необходимо до выполнения расчетов поместить числовые данные в области памяти a и b. Для указания операции ввода данных в схеме используется блок 2. Процесс получения результата вычислений описывается в блоке 3.



	Рис.2.  Блок-схема алгоритма 

решения задачи 1
	


Поскольку результат вычисления заданного выражения находится в области Y оперативной памяти, то необходимо использовать вывод информации на экран дисплея для восприятия выходных данных человеком. Описание процесса вывода информации дается в блоке 4.
Для составления программы решения задачи 1 рассмотрим несколько операторов языка QBASIC, которые могут использоваться для программирования линейных вычислительных процессов.

Оператор INPUT

INPUT  – оператор ввода данных с клавиатуры.

Полный формат этого оператора :

       INPUT [;] [“приглашение”  {;|,}] список переменных


                                                                      ; добавляет знак вопроса
                                                                      в строку приглашения
                                                      необязательная строка, отображаемая 
                                                      на экране перед вводом данных

                                           ; оставляет курсор в той же строке при нажатии <Enter>

В списке переменных перечисляются имена переменных, которым присваиваются вводимые значения. Имена переменных в списке разделяются запятыми.

Примечание. Все необязательные компоненты в структуре оператора заключаются в квадратные скобки. Альтернативные элементы заключаются в фигурные скобки и разделяются между собой символом |.

Как видно из формата оператора, обязательными его компонентами являются только: INPUT список переменных

Примеры использования оператора:

                                 INPUT a,b

                                 INPUT  “ Введите два числа” ; a,b

Оператор присваивания
Формат оператора присваивания имеет следующий вид:

[LET] переменная = выражение


                                                           Символ присваивания

Оператор присваивания вычисляет выражение, расположенное справа от символа присваивания и результат вычисления присваивает переменной, расположенной слева от символа присваивания. Ключевое слово оператора LET является необязательным.

Примеры использования оператора присваивания:

                                           y= (a+b)^3+m

                                           x=0

                                           a=b

Оператор PRINT

Для вывода данных на экран используется оператор PRINT в следующем формате:

PRINT [выражение] [ {; | , } ]

Использование оператора PRINT без параметров приводит к выводу пустой строки (перемещению курсора на следующую строку). Здесь выражение – это одно или несколько символьных или числовых выражений. Если выражений в одном операторе несколько, то они разделяются либо запятой, либо точкой с запятой.

При использовании запятой в качестве разделителя вывод следующего после запятой выражения осуществляется с начала следующей зоны вывода.

Примечание. Экран разделяется на пять зон вывода по 14 символов в каждой зоне.

При использовании точки с запятой последующее выражение выводится рядом с предыдущим.

Примеры использования оператора:

                                           PRINT y

                                           PRINT “Результат=“;у

Оператор END
END – оператор окончания программы, записывается последним в тексте программы.

Составим программу решения задачи 1 путем кодирования блок -схемы (рис.2) операторами QBASIC, только что изученными нами:

REM программа решения  задачи 1

INPUT a,b            ‘ соответствует блоку ввода, список переменных указан внутри блока 2

y= (a^2+b^2)/100  ’ оператор присваивания соответствует блоку “процесс”

PRINT “y=“;y         ‘ выводимые значения указаны внутри блока вывода 

END
2.2. Ветвящиеся вычислительные процессы

Ветвящимся называется такой вычислительный процесс, в котором выбор направления обработки информации зависит от исходных или промежуточных данных (от результатов проверки выполнения какого-либо логического условия).

В качестве примера ветвящейся алгоритмической структуры рассмотрим процесс вычисления выражения задачи 2: 

                                                     


Появление условия при решении этой задачи связано с возможным делением на ноль. Такая ситуация возникает, если будут введены в области памяти a и b два одинаковых числа.

Блок-схема решения задачи 2 показана на рис.3.

Рассмотрим особенности построения этой схемы алгоритма. Блоки 3,4,5,6 представляют единую конструкцию «альтернатива». Начинается эта конструкция с блока 3 (блока «Решение»), из которого выходят две ветви алгоритма (два плеча альтернативы), определяющие отдельные направления обработки информации. 

       a, b

                          Да      3                   Нет

      a - b ( 0

                 4                                     6    

	Блоки 4 и 5 расположены на ветви «ДА», а блок 6 – на ветви «НЕТ». Для данной алгоритмической структуры характерно, что в любой момент ее реализации осуществляется обработка только по какой-либо одной из ветвей.

Для описания ветвящегося вычислительного процесса ранее рассмотренную группу операторов пополним еще одним.



	Рис.3. Блок-схема алгоритма
      решения задачи 2
	


Условный оператор IF
Для описания конструкции «альтернатива» используется условный оператор блочной или линейной структуры.

Формат оператора IF блочной структуры:

IF  условие  THEN

блок-операторов1

[ ELSE

блок-операторов2 ]

END  IF

Формат оператора IF линейной структуры:

IF  условие  THEN операторы1 [ ELSE операторы2 ]

Обе структуры  оператора могут быть использованы как в полной форме :

 IF ... THEN ... ELSE   , так и в усеченной форме: IF ... THEN . Выбор формы оператора IF зависит от наполнения ветвей альтернативы. Если по ветви «НЕТ» действия определены (как в схеме на рис.3), то используется полная форма оператора, а если такие действия не определены («пустая» ветвь), то используется усеченная форма.

Рассмотрим работу оператора IF. Сначала проверяется выполнение условия. Если условие выполняется, то реализуются действия, описанные как блок-операторов1 ( или операторы1), в противном случае – действия указанные в блок-операторов2 (или операторы2).

Как блок-операторов1 ,так и блок-операторов2 представляют собой один или несколько операторов в одной или нескольких строках.
Условие – это логическое выражение, результатом выполнения которого могут быть только два значения: истинное (не ноль) или ложное (ноль).

Для построения логических выражений могут быть использованы операции сравнения и логические операции.

Известны шесть операций сравнения : = (равно); > (больше);
 < (меньше); <> (не равно); <= (меньше или равно); 
>= (больше или равно).

Операции сравнения имеют более низкий приоритет, чем арифметические операции. Логические операции используются для построения сложных логических выражений и будут рассмотрены нами в разделе 5.

Запишем текст программы решения задачи 2, используя схему алгоритма на рис.3 и условный оператор  IF блочной структуры. 

REM  программа решения  задачи 2 (вариант 1)

INPUT a , b

IF a – b <> 0 THEN

y = (a ^ 2 + b ^ 2 ) / ( a – b )

PRINT  y

ELSE

PRINT “Решения нет “

END IF

END

Текст программы решения этой же задачи с использованием линейной структуры оператора IF имеет следующий вид:

REM  программа решения  задачи 2 (вариант 2)

INPUT a , b

IF a - b <> 0 THEN y = (a ^ 2 + b ^ 2 ) / ( a -– b ):PRINT  y ELSE PRINT “Решения нет” 

END

Можно использовать расширенную форму записи оператора IF с форматом:

IF условие1 THEN

блок-операторов1

ELSEIF условие2 THEN

блок-операторов2

. . .

ELSE

блок-операторовn
END IF

Блоков ELSEIF может быть использовано неограниченное количество.

Рассмотрим пример использования расширенной формы оператора IF при решении задачи 3, в которой требуется вычислить Y:


        X2 – 10, если X<0
Y =  18,        если X=0

         X3 + 5,   если X>0

REM программа решения задачи 3

INPUT x

IF x< 0 THEN

y = x ^2–10

ELSEIF x = 0 THEN

y = 18

ELSE

y = x^3 + 5

END IF

PRINT y

END

Оператор выбора SELECT CASE
Формат оператора:

SELECT CASE выражение
CASE условие1

блок-операторов1

CASE условие2

блок-операторов2

. . . . . . . . . . . .

[CASE ELSE

блок-операторовn]

END SELECT

Оператор проверяет условия и определяет оператор, который будет выполняться. Значение выражения используется для проверки в условиях. Число CASE-блоков внутри оператора SELECT CASE не ограничено. Блок CASE ELSE получает управление, если ни одно из условий не выполняется. Здесь выражение – это любое допустимое выражение, значение которого – число или строка. Значение этого выражения проверяется в списках условий, каждое из которых может быть записано в одной из трех допустимых форм:

· список выражений. Проверяемое значение должно равняться значению одного из перечисленных через запятую выражений;

· проверяемое значение должно укладываться в диапазон значений: <выражение1> TO <выражение2>;

· CASE IS <операция сравнения> <проверочное выражение>
Слово IS обозначает значение выражения.

Использование оператора SELECT CASE иллюстрируется в следующем тексте программы:

REM использование оператора выбора
INPUT  a 
SELECT  CASE  a ‘указывается проверяемое выражение а 

CASE  1, 2, 3 ‘значение а сравнивается с константами 1, 2 и 3

PRINT " Значение a равно  1, 2 или 3"

CASE  4  TO  10 ‘проверяется, находится ли значение a  в интервале от 4 до 10

PRINT "Значение a находится в диапазоне от 4 до 10"

CASE  IS > 10 ‘проверяется, удовлетворяет ли значение а условию a> 10

PRINT " Значение a больше 10"

END SELECT

END

2.3. Циклические вычислительные процессы

Циклом называется многократно повторяемый участок вычислений. Классификация циклов представлена на рис.4.

                                                 Циклы


       с определенным числом             с неопределенным числом
                повторений                                     повторений

       ( с параметром )

                                                   с предусловием           с постусловием

Рис.4. Классификация циклов

2.3.1. Цикл с определенным числом повторений (c параметром)

Рассмотрим принцип работы цикла с параметром на примере задачи 4.

Задача 4. Получить результаты расчетов по формуле 

                 

  при значениях   – 5 <= a <= 5 с шагом  +1

Блок-схема решения задачи 4 представлена на рис.5.

	На схеме можно выделить компоненты, характерные для цикла с параметром. К таким компонентам относятся:

1) наличие стрелки возврата 
(к блоку 4);

2) наличие трех характерных блоков :

· блок (Процесс), в котором параметр цикла принимает начальное значение а= - 5 
(в блоке 3);

· блок (Процесс), в котором параметр цикла изменяет свое значение на определенную величину a = a +1  (в блоке 6);

· блок (Решение), в котором параметр цикла сравнивается с последним возможным значением 5  (в блоке 7).



	Рис. 5. Блок-схема алгоритма

      решения задачи 3.


	


Три блока (3,6,7) называются заголовком цикла, а блоки (4 и 5), расположенные между блоками заголовка цикла, образуют тело цикла.

Общий вид управляющей структурированной конструкции “цикл с параметром” приводится на рис.6.

Для описания циклического вычислительного процесса данного вида QBASIC имеет специальный оператор цикла с параметром.

Общий вид управляющей структурированной конструкции “цикл с параметром” приводится на рис.6.

Для описания циклического вычислительного процесса данного вида QBASIC имеет специальный оператор цикла с параметром.




имя = начало


операторы 
тела цикла

имя = имя + шаг

                                        Да

имя ≤ конец

                 Нет


Рис.6. Управляющая конструкция «цикл с параметром»
Оператор цикла с параметром FOR ...NEXT
Формат оператора цикла с параметром:

FOR переменная = начало TO конец  [ STEP приращение]

операторы тела цикла

NEXT [переменная]

Здесь переменная  представляет собой числовую переменную, используемую как параметр цикла; 

начало, конец   - это начальное и конечное значения параметра цикла;

приращение - величина, на которую изменяется параметр на каждом шаге цикла. Приращение может иметь как положительное, так и отрицательное значение. В случае, когда приращение равно +1, конструкция STEP  приращение  может отсутствовать в операторе.

При положительном приращении оператор цикла с параметром выполняется только в том случае, если начало плюс приращение меньше или равно значению конец параметра цикла. Выход из цикла происходит, когда значение параметра цикла становится равно значению конец плюс приращение.

Примечание. Досрочный выход из цикла  FOR ... NEXT  может быть реализован оператором EXIT FOR.

Напишем текст программы решения задачи 4, пользуясь ранее составленной блок-схемой (рис.5).

REM  программа решения задачи 4

INPUT b

FOR a = – 5 TO 5 STEP 1  ‘ соответствует блокам заголовка цикла
y = ( a + b ) ^2 / 1000      ‘ оператор тела цикла

PRINT y                           ‘ оператор тела цикла
NEXT a

END

2.3.2. Цикл с предусловием 

Рассмотрим использование цикла с предусловием при решении задачи 5, в которой требуется вывести все значения x >1 , причем каждое последующее значение x  получается при делении предыдущего пополам.

                      x

             5 

                x = x / 2

                
	                    Да         Условие     Нет 

	Рис.7.  Блок-схема алгоритма решения задачи 5 с использованием цикла
 с предусловием
	Рис.8. Управляющая конструкция:

«цикл с предусловием»


Схема решения этой задачи приведена на рис.7. На этой схеме можно выделить условие, остающееся истинным при выполнении цикла (блок 3).

Такое условие называется  инвариантом цикла. Блоки 4 и 5 представляют тело цикла. Управляющая конструкция «цикл с предусловием» приведена на рис. 8. 

Цикл с предусловием  является циклом «пока» и в ряде случаев может быть не выполнен ни разу, что должно соответствовать задуманному алгоритму.

Так, например, если при решении задачи 5 (см. рис.7) в качестве начального значения x  мы введем значение 0.9, то тело цикла (блоки 4 и 5) не выполнится ни разу.

Цикл с предусловием в QBASIC может быть реализован двумя различными операторами: WHILE ... WEND  и DO ...  LOOP с проверкой условия в начале.

Оператор цикла WHILE ... WEND
Формат оператора:

WHILE  условие выполнения цикла

операторы тела цикла

WEND
Операторы тела цикла выполняются до тех пор, пока указанное условие истинно. Условие представляет собой логическое выражение, о правилах построения которого было рассказано выше.

Напишем программу решения задачи 5 на основе блок-схемы (рис.7) и с использованием оператора цикла WHILE ... WEND.

REM  программа решения задачи 5 (вариант 1)

INPUT x

WHILE  x > 1

PRINT x

x = x / 2

WEND

END

Оператор цикла DO ... LOOP  с проверкой условия в начале
Формат записи этого оператора:

DO { WHILE условие выполнения | UNTIL условие прекращения }

Операторы тела цикла

LOOP

Как видно из приведенной формы оператора, в начале цикла может проверяться условие выполнения цикла (при использовании ключевого слова WHILE) или условие прекращения цикла (при использовании ключевого слова UNTIL). Оператор DO … LOOP с проверкой цикла в начале повторяет операторы тела цикла, пока условие выполнения цикла истинно или условие прекращения цикла ложно.

Напишем текст программы решения задачи 5 на основе блок-схемы (рис.7), используя оператор цикла с предусловием 
 DO WHILE ... LOOP.

REM  программа решения задачи 5 (вариант 2)

INPUT x

DO WHILE x > 1

PRINT x

x = x / 2

LOOP

END

Текст программы решения этой же задачи с использованием оператора цикла с предусловием DO UNTIL ... LOOP будет следующим:

REM  Программа решения задачи 5 (вариант 3)

INPUT x

DO UNTIL x <= 1

PRINT x

x = x / 2

LOOP

END

2.3.3. Цикл с постусловием

Рассмотрим использование цикла с постусловием при решении предыдущей задачи 5.

Схема решения этой задачи приведена на рис.9. На этой схеме можно выделить условие, остающееся истинным при выполнении цикла (блок 5), инвариант цикла. Блоки 3 и 4 представляют тело цикла. Управляющая конструкция «цикл с постусловием» приведена на рис.10. 

Цикл с постусловием является циклом «до» и отличается от рассмотренных ранее видов циклов тем, что должен выполниться хотя бы один раз.

Так, например, если при решении задачи 5 (см. рис.9) в качестве начального значения x мы введем значение 0.9, то тело цикла (3 и 4 блоки) выполнится один раз обязательно.

Цикл с постусловием в QBASIC может быть реализован оператором DO ...  LOOP с проверкой условия в конце.

Оператор цикла DO ... LOOP с проверкой условия в конце

Формат записи этого оператора:

DO

Операторы тела цикла

LOOP { WHILE условие выполнения | UNTIL условие прекращения }

	Рис.9. Блок-схема алгоритма решения задачи 5 с использованием цикла
 с постусловием
	Рис.10. Управляющая 

конструкция:

«цикл с постусловием»


Как видно из приведенной формы оператора, в конце цикла может проверяться условие выполнения цикла (при использовании ключевого слова WHILE) или условие прекращения цикла (при использовании ключевого слова UNTIL). Оператор DO … LOOP с проверкой условия в конце повторяет операторы тела цикла, пока условие выполнения цикла истинно или условие прекращения цикла ложно.

Напишем текст программы решения задачи 5 на основе блок-схемы (см. рис.9) с использованием оператора цикла с постусловием DO ...  LOOP   WHILE.
REM  программа решения задачи 5 (вариант 4)

INPUT x

DO

PRINT x

x = x / 2

LOOP WHILE x > 1

END

Текст программы решения этой же задачи с использованием оператора цикла с постусловием DO ... LOOP  UNTIL будет следующим:

REM  программа решения задачи 5 (вариант 5)

INPUT x

DO

PRINT x

x = x / 2

LOOP UNTIL x <= 1

END

Примечание. Досрочный выход из оператора цикла DO ... LOOP может быть организован с помощью оператора EXIT...DO

3. АЛГОРИТМИЗАЦИЯ И ПРОГРАММИРОВАНИЕ НА QBASIC ЗАДАЧ ОБРАБОТКИ МАССИВОВ

3.1. Общие сведения об обработке массивов

Массивом называется совокупность элементов с одинаковыми свойствами. Любой массив характеризуется :

· именем;
· размерностью;
· типом элементов.
Каждый элемент массива имеет определенное значение, координаты своего местоположения в массиве и обозначается при помощи переменной с индексами. Количество индексов зависит от размерности массива.

Массивы могут быть одномерные, двумерные и т. д. В данном разделе остановимся на изучении массивов числового вещественного типа обычной точности.

Обработка любого массива представляет собой циклический вычислительный процесс (как правило, цикл с параметром), в котором параметрами циклов яв​ляются индексы, а в теле циклов используются выражения с индексированными переменными.

Основными видами задач обработки массивов являются :

· определение суммы значений элементов, произведения значений элементов и среднего арифметического для всех элементов массива;

· определение суммы значений, произведения значений, количества элементов и среднего арифметического для элементов массива, удовлетворяющих определенным условиям;

· определение максимального (минимального) по значению элемента и его местоположения в массиве;

· упорядочение значений элементов в массиве.

В программе массивы до их использования должны быть объявлены оператором DIM.

Оператор DIM

Формат оператора:

DIM имя-массива (границы) [, имя-массива (границы) ... ]

Оператор DIM выделяет в памяти ПК для массивов необходимый объем. Области памяти, выделенные для элементов числовых массивов при этом обнуляются (очищаются).

Примечание. Массив в QBASIC может быть задан неявно. Если при написании программы используется элемент массива без предварительного объявления этого массива, например, неявно заданный одномерный массив содержит 10 элементов.

3.2. Одномерные массивы

Одномерный массив носит название вектора. Элементы одномерного массива имеют по одному индексу. Этот индекс соответствует номеру элемента в векторе.

Рассмотрим вектор  A, состоящий из 7 элементов со значениями: 30, 25, 18, 20, 7, 11, 9. Любой элемент этого вектора обозначается  A( i ) , где i – индекс,      1 <= i <= 7.

При i = 1  A( i ) = 30      или  A( 1 )= 30;

при  i = 5  A ( i ) = 7  или A ( 5 ) = 7.

Элементы одномерного массива располагаются один за другим в последова​тельно расположенных областях памяти. Рассмотрим типовые задачи обработки вектора.

Задача 6. Определить и вывести сумму значений элементов в массиве A , содержащем 7 элементов.

Рис.11. Блок-схема алгоритма 
              решения задачи 6 


	Блок-схема алгоритма решения данной задачи представлена на рис.11. Как видно из схемы, процесс решения поставленной задачи включает в себя два последовательно расположенных цикла с параметром. 

Блоки 2 – 5 описывают циклический процесс ввода элементов одномерного массива в память. Блоки 7 - 10 предназначены для организации цикла накопления суммы элементов массива «нарастающим итогом». При решении задач подсчета суммы значений элементов массива определяется «чистая» область памяти, в которой должна накапливаться сумма (блок 6).

Примечание. В QBASIC все числовые переменные имеют начальное нулевое значение, поэтому в схеме на рис.11 блок 6 может быть опущен.

Составим программу решения 
задачи 6 на основе блок-схемы, представленной на рис.11.

REM программа решения задачи 6

DIM A ( 7 )

FOR  i = 1  TO   7 

INPUT  A( i )

NEXT  i

S=0

FOR  i = 1  TO  7

S =  S + A ( i )

NEXT  i

PRINT  S

END

P=P*  V(i)      

K = K + 1
	Задача 7. Определить количество и произведение значений отрицательных элементов в векторе. Как видно, постановка задачи дана в общем виде. 

Блок-схема алго​ритма решения такой задачи приведена на рис.12. В блоке 2 осуществляется ввод количества элементов массива (в переменную n ). Блоки 3–6 описывают ввод в цикле n элементов массива с произвольно заданным именем V. В блоке 7 подготавливается область памяти для подсчета произведения значе​ний элементов ( p = 1), а в блоке 8 – для подсчета количества элементов(k=0 ).

Блоки 9–14 организуют циклический процесс подсчета количества и произведения значений отрицательных элементов. Программа, составленная в соответствии с алгоритмом решения задачи 7, бу​дет иметь следующий вид :
REM программа решения задачи 7
INPUT n

DIM V ( n )

FOR  i = 1  TO  n

INPUT V ( i )

NEXT i

p = 1:k = 0

FOR  i = 1  TO  n

IF  V (i) < 0  THEN

p = p * V ( i )

k = k + 1

END  IF

NEXT i

PRINT p ; k: END

	Рис.12.  Блок-схема алгоритма к задаче 7

	Задача 8. В векторе K, содержащем 12 элементов, определить значение максимального элемента и его номер.

Блок-схема алгоритма решения этой задачи показана на рис.13. Как видно, при решении этой задачи подготавливается область M, в которой должно находиться наибольшее (на каждом шаге цикла) значение элемента вектора. Переменная  N запоминает номер того элемента вектора, значение которого хранится на данном шаге цикла в области M. Первоначально переменная M получает значение первого элемента вектора (блок 6), а переменная 
N – соответственно номер первого элемента вектора (блок 7). В последующем при поочередном выборе всех значений элементов вектора организуется сравнение каждого из этих значений с содержимым переменной M. В случае, если значение выбранного из вектора элемента оказывается больше сравниваемого с ним значения, записанного в M, то переменной M присваивается значение нового выбранного элемента, а переменной N – номер этого элемента.

Программа решения 
задачи 8 на QBASIC имеет следующий вид:



	Рис.13.  Блок-схема алгоритма

решения задачи 8

	REM программа решения задачи 8

DIM K ( 12 )

FOR  i = 1  TO   12 

INPUT K ( i )

NEXT i

M = K ( 1 ) : N = 1

FOR  i = 1  TO  12

IF K ( i ) > M  THEN 

M = K ( i ) : N = i 

END IF

NEXT  i

PRINT  M, N

END

Задача 9. Упорядочить по возрастанию значений элементы вектора.

Блок-схема решения задачи приведена на рис.14. Так как задача поставлена в общем виде, то сначала организуется ввод количества элементов вектора в переменную m (блок 2), а потом уже в цикле ввод всех  m элементов вектора. Предложенная схема реализует так называемый метод «пузырька». Этот метод основывается на попарном сравнении близлежащих элементов вектора (блок 9). Сравниваться должны все элементы вектора. Если предыдущий элемент оказывается больше последующего, то эти элементы обмениваются местами. Однако за один сеанс сравнения всех элементов вектора и возможной их перестановки данный метод может не привести к желаемому результату, поэтому процедуру сравнений и перестановок необходимо возобновить.

	Рис.14.  Блок-схема алгоритма 

решения задачи 9
	


Количество таких сеансов обработки всех элементов вектора, приводящее обязательно к его полной упорядоченности, равно числу элементов вектора минус  1 ( m – 1), поэтому целесообразно использовать возможность сокращения числа сеансов обработки вектора. Сигналом к тому, что вектор отсортирован, будем считать  факт отсутствия перестановок при сравнении элементов. Для получения информации о наличии перестановок после очередного сеанса обработки всех элементов вектора используется специальная переменная ( в схеме на рис. 14 пере​менная N ). До начала сравнения элементов эта область принимает опреде​ленное значение 
(N = 0 в блоке 7). В случае наличия хотя бы одной переста​новки значение этой переменной изменяется ( N = 1 в блоке 13). По окончании очередного просмотра всех элементов вектора проверяется значение перемен​ной N. Если оно отлично от первоначально заданного, то массив нельзя считать отсортированным и процедура сортировки возобновляется. Если значение об​ласти N  не изменилось в процессе сравнения  элементов, то процедуру сорти​ровки массива можно считать завершенной.
Обратите внимание на процесс обмена значениями двух элементов вектора (блоки 10 – 12). Во избежание потери их значений используется дополнительная переменная. 
Составим текст программы решения задачи 9 в соответствии с блок-схемой на рис. 14.

REM программа решения задачи 9 (вариант 1)

INPUT  m

DIM  C ( m )

FOR  i = 1  TO  m

INPUT   C ( i )

NEXT  i

DO

N =0

FOR  i = 1  TO  m-1

IF  C ( i ) >  C ( i+1 )  THEN

B = C ( i ) : C ( i ) = C ( i+1 ): C ( i + 1 ) = B : N = 1

END IF

NEXT i

LOOP  WHILE  N = 1

FOR  i = 1  TO  m

PRINT  C ( i )

NEXT  i

END

Алгоритм этой задачи может быть реализован с использованием других операторов QBASIC. Рассмотрим дополнительно операторы SWAP и GOTO.

Оператор SWAP

Этот оператор используется для обмена значениями двух переменных.

Формат оператора:

                          SWAP переменная1 , переменная2

После выполнения этого оператора переменная1 получает значение переменной2 и наоборот.

Оператор GOTO

Используется для передачи управления на определенный участок программы.

Оператор записывается в виде:

GOTO метка

Здесь метка  – это числовая или символьная метка той строки программы, на которую передается управление.

Например:        GOTO 10

                          GOTO  m1

Примечание. Оператор GOTO – старейший и неструктурированный оператор языка BASIC. Программа, написанная с использованием этого оператора, трудно читается, отлаживается и модифицируется, так как не имеет четкой структуры. Не существует таких программных конструкций, где действительно было бы необходимо применение оператора GOTO.
Познакомьтесь с другим вариантом написания программы решения задачи 9, включающим использование только что изученных операторов.

REM программа решения задачи 9 ( вариант 2)

INPUT  m

DIM  C ( m )

FOR  i = 1  TO  m

INPUT   C ( i )

NEXT  i

met: N =0

FOR  i = 1  TO  m-1

IF  C ( i ) >  C ( i+1 )  THEN

SWAP  C ( i ) , C ( i+1 ): N = 1

END IF

NEXT i

IF N = 1 THEN GOTO met

FOR  i = 1  TO  m

PRINT  C ( i )

NEXT  i
END

3.3. Двухмерные массивы

Двухмерный массив носит название матрицы. Рассмотрим числовую матрицу B, состоящую из 4 строк и 3 столбцов (рис. 15).

Рис.15. Пример числовой матрицы, состоящей из 4 строк и 3 столбцов

Расположение элемента в двухмерном массиве определяется номером строки и номером столбца, на пересечении которых находится этот элемент, поэтому  каждый элемент матрицы имеет два индекса: первый индекс указывает на номер строки, а второй индекс показывает номер столбца. 

Если номер строки обозначить буквой i, а номер столбца - буквой j, то для рассматриваемой нами матрицы B (см. рис. 15) будут справедливы следующие утверждения:

при i = 1 и j = 2  B(i,j)= 2;  при i = 3 и j = 1  B(i,j)= 10 и т. д.

Рассмотрим типовые задачи обработки двухмерных массивов.
Задача 10. Определить и вывести среднее арифметическое для элементов матрицы B, состоящей из 4 строк и 3 столбцов.

Среднее арифметическое представляет собой отношение суммы значений элементов к количеству этих элементов. Количество в данной задаче определять не нужно, так как оно известно (4•3), поэтому основная обработка сводится к определению суммы значений элементов.

Блок-схема алгоритма решения этой задачи приведена на рис.16.

На схеме хорошо видны два последовательно расположенных циклических участка: один – для организации ввода данных 
(блоки 2– 8), другой – для организации вычисления суммы значений элементов матрицы (блоки  10 -16).

Рис.16. Блок-схема алгоритма решения задачи 10

	Каждый из этих циклических участков представляет собой вложенные циклы.

Вложенными  называются циклы, расположенные один в другом. Цикл, яв​ляющийся внешним, включает в свое тело внутренний цикл полностью.

Так, на схеме (рис. 16) можно различить заголовок внешнего цикла c параметром i 
(блоки 2,7,8), тело внешнего цикла 
(блоки 3 - 6), которое представляет собой внутренний цикл с параметром j со своим заголовком (блоки 3, 5, 6) и телом цикла (блок 4).

Текст программы, реализующей на QBASIC представленный на рис.16 вычислительный процесс, имеет следующий вид:

REM программа решения задачи 10

DIM  B( 4, 3 )

FOR  i = 1  TO  4

FOR  j = 1  TO  3

INPUT  B( i , j )

NEXT  j

NEXT  i

S=0

FOR  i = 1  TO  4

FOR  j = 1  TO  3

S = S + B ( i , j )

NEXT  j

NEXT  i

S = S / 12

PRINT  S

END

Задача 11. В матрице D, состоящей из  4 строк и 5 столбцов, определить среднее арифметическое среди положительных элементов и среднее арифметическое среди отрицательных элементов.

Для решения данной задачи необходимо определить сумму и количество как положительных, так и отрицательных элементов, а при вычислении средних арифметических следует убедиться в наличии как положительных, так и отрицательных чисел в матрице.

	Рис.17. Блок-схема алгоритма решения задачи 11


Блок - схема алгоритма решения данной задачи показана на рис.17. 

Приведем текст программы решения задачи 11.

REM программа решения задачи 11

DIM  D( 4, 5 )

FOR  i = 1  TO  4

FOR  j = 1  TO  5

INPUT  D( i , j )

NEXT  j

NEXT  i

S1 = 0: K1 = 0: S2 = 0: K2 = 0

FOR  i = 1  TO  4

FOR  j = 1  TO  3

IF  D ( i , j ) > 0 THEN

S1=S1 + D ( i, j ): K1 = K1 + 1

ELSE

IF  D ( i , j ) < 0 THEN

S2 = S2 + D ( i , j ): K2 = K2 + 1

END  IF

END  IF

NEXT  j

NEXT  i

IF  K1 <> 0  THEN

SP = S1 / K1

PRINT  SP

ELSE

PRINT  “Нет положительных”

END IF 

IF  K2 <> 0  THEN

SO = S2 / K2

PRINT  SO

ELSE

PRINT  “Нет отрицательных”

END IF 

END

Задача 12. В произвольной матрице определить значение минимального элемента и его координаты (номер строки и номер 
столбца).

Для определения произвольной матрицы необходимо организовать по запросу ввод количества строк и столбцов матрицы, дать матрице произвольное имя и объявить в операторе DIM.

Блок-схема алгоритма решения этой задачи показана 
на рис. 18.

Рис.18. Блок-схема алгоритма
решения задачи 12

Текст программы решения задачи имеет следующий вид:

REM программа решения задачи 12

INPUT n , k 

DIM  A( n, k )

FOR  i = 1  TO  n

FOR  j = 1  TO  k

INPUT  A( i , j )

NEXT  j

NEXT  i

M = A ( 1 , 1 ) : T = 1 : L = 1

FOR  i = 1  TO  n

FOR  j = 1  TO  k

IF A ( i , j ) < M THEN

M = A ( i, j ) : T = i : L = j

END  IF

NEXT  j

NEXT  i

PRINT  M ; T ; L

END

Рис.19. Блок-схема алгоритма 
решения задачи 13

Задача 13. Упорядочить по убыванию значений элементы в каждой строке матрицы С, состоящей из 5 строк и 4 столбцов.

Проведем сортировку методом «пузырька» элементов каждой строки матрицы отдельно. К сортировке каждой последующей строки будем приступать только после полного упорядочения элементов предыдущей строки.

Блок-схема решения задачи 13 показана на рис.19.

Приведем текст программы решения данной задачи.




REM программа решения задачи 13

DIM   C ( 5 , 4 )

FOR  i = 1  TO  5

FOR  j = 1  TO  4

INPUT  C( i , j )

NEXT  j

NEXT  i

FOR  i = 1  TO  5

DO 

N = 0

FOR  j = 1  TO  3

IF C ( i , j ) < C ( i , j+1 ) THEN

B = C ( i , j ): C ( i , j ) = C ( i , j+1) : C ( i , j+1) = B : N = 1

END  IF

NEXT  j

LOOP WHILE  N = 1

NEXT  i

FOR  i = 1  TO  5: FOR  j = 1  TO  4

PRINT  C( i , j )

NEXT  j:NEXT  i:END

Задача 14. В матрице A , состоящей из 4 строк и 3 столбцов, определить номер строки, содержащей наибольшую сумму значений элементов.

Для решения этой задачи целесообразно определить и использовать вспомогательный  вектор c количеством элементов, совпадающим с количеством строк матрицы. Сумма значений каждой строки матрицы должна накапливаться в соответствующем элементе вспомогательного вектора (рис. 20).





                  Матрица                   Вектор

Рис.20. Использование вспомогательного вектора

После окончания процесса накопления суммы значений элементов каждой строки матрицы в элементах вспомогательного вектора необходимо определить номер максимального элемента в векторе, так как он будет совпадать с номером искомой строки матрицы. Блок - схема решения этой задачи приведена на рис. 21.

Рис.21. Блок-схема алгоритма 
решения задачи 14
Программа решения данной задачи имеет следующий вид:

REM программа решения задачи 14

DIM   A ( 4 , 3 ) ,S (4 )

FOR  i = 1  TO  4

FOR  j = 1  TO  3

INPUT  A( i , j )

NEXT  j

NEXT  i

FOR  i = 1  TO  4

S ( i ) = 0

FOR  j = 1  TO  3

S ( i ) = S ( i ) + A ( i , j )

NEXT  j

NEXT  i

M = S ( 1 ) : N = 1

FOR  i = 1  TO  4

IF  S ( i ) > M  THEN

M = S ( i ) : N = i

END IF

NEXT  i

PRINT  N

END


4. ЭЛЕМЕНТЫ МОДУЛЬНОГО ПРОГРАММИРОВАНИЯ В QBASIC

4.1. Общие сведения о программных модулях

Программным модулем называется программа, используемая в качестве составной части других программ. Каждая программа на QBASIC может состоять из одного или нескольких модулей, один из которых является главным, а другие – вспомогательными.

Обычно использование программного модуля предполагает наличие операторов для описания программного модуля и оператора для вызова программного модуля на выполнение.

При наличии описания программный модуль может быть выполнен только в том случае, если он вызывается на выполнение. При отсутствии описания программный модуль выполнен быть не может.

QBASIC может подключать для выполнения только внутренние модули, определенные в главной программе.
В QBASIC в отличие от традиционных версий BASIC имеется развитый аппарат описания подпрограмм и функций с параметрами, что позволяет создавать сложные модульные программы. К таким средствам относятся процедуры SUB и FUNCTION.

4.2. Процедуры SUB и FUNCTION

Для каждой процедуры  SUB  и FUNCTION, используемой в программе, QBASIC генерирует соответствующий оператор DECLARE для её объявления.

Формат оператора DECLARE :


DECLARE    SUB|FUNCTION    имя-процедуры [(список параметров)]

Процедура SUB

Описание процедуры SUB имеет следующий вид:

SUB имя [(список формальных параметров)]

операторы

END SUB

Вызов процедуры SUB может быть осуществлен двумя способами:

1)с использованием специального ключевого слова («фортрановский вызов»):

CALL имя [( список фактических параметров)]

2) без использования специального ключевого слова («паскалевский вызов»):

имя [ список фактических параметров]

Фактические параметры представляют собой передаваемые процедуре значения.

Формальные параметры используются в процедуре для хранения переданных фактических параметров.

При вызове процедур обязательно совпадение количества и типов формальных и фактических параметров.

Решим задачу с использованием процедуры SUB.

Задача 15. Подсчитать сумму значений элементов трех одномерных массивов a (10), b (8) , c (6).

Хотя имена и количество элементов данных векторов различны, алгоритмы обработки для каждого вектора одинаковы: каждый вектор нужно ввести в оперативную память и для каждого вектора необходимо вычислить сумму значений.

	Рис.22. Система блок-схем алгоритмов решения задачи 15


Для описания вычислительного процесса поставленной задачи построим си​стему блок- схем алгоритмов (рис. 22). Система включает блок-схему алго​ритма основной программы а), блок-схему алгоритма подпрограммы ввода элементов вектора b) и блок-схему алгоритма подпрограммы вычисления суммы значений элементов вектора c). В блок-схему алгоритма основной программы включены блоки «Предопределенный процесс» (блоки 2 – 7) для указания трех вызовов подпрограммы vv, осуществляющей  ввод элементов каждого вектора, и трех вызовов подпрограммы sv, вычисляющей сумму значений элементов вектора. При вызове указываются фактические параметры, передаваемые программе. В случае передачи параметра-массива после имени массива записывают пустые круглые скобки (). Блок-схемы на рис. b) и c) представляют собой описание подпрограмм, вызы​ваемых из основной программы. Все формальные параметры, указанные при описании подпрограмм, при их вызове будут замещены фактическими параметрами.

Приведем текст программы  решения задачи 15:

DECLARE SUB vv (d!(), n!)

DECLARE SUB sv (d!(), n!, sd!)

REM программа решения задачи 15 (вариант 1)

DIM a(10), b(8), c(6)

CALL vv(a(), 10)

CALL vv(b(), 8)

CALL vv(c(), 6)

CALL sv(a(), 10, sa)

CALL sv(b(), 8, sb)

CALL sv(c(), 6, sc)

s = sa + sb + sc

PRINT s

END

SUB sv (d(), n, sd)

sd = 0

FOR i = 1 TO n

sd = sd + d(i)

NEXT i

END SUB

SUB vv (d(), n)

FOR i = 1 TO n

INPUT d(i)

NEXT i

END SUB

Процедура FUNCTION

Описание процедуры FUNCTION имеет следующий вид :

FUNCTION имя [ ( формальные параметры )]

[ операторы]

имя  =  выражение

END  FUNCTION

Здесь имя задается как обычная переменная и определяет тип возвращаемых данных; выражение – возвращаемое значение функции. Если имени функции ничего не присваивается, возвращается значение по умолчанию: для числовых функций – ноль, для символьных – пустая строка ("").
Вызывается  FUNCTION  подобно встроенной функции  языка QBASIC.

переменная  =  имя [(фактические параметры )]

Примечание. Для вызова функции ее имя со списком фактических параметров может использоваться везде, где может использоваться выражение.

Используем для решения задачи 15 функцию fv, вычисляющую сумму значений элементов вектора:

DECLARE SUB vv (d!(), n!)

DECLARE FUNCTION fv! (d!(), n!)

REM программа решения задачи 15(вариант 2)

DIM a(10), b(8), c(6)

vv a(), 10                 ‘ "паскалевский вызов” процедуры SUB

vv b(), 8

vv c(), 6

sa = fv(a(), 10)                  Эти четыре оператора 

sb = fv(b(), 8)                     можно заменить одним
sc = fv(c(), 6)                     S = fv ( a ( ), 10 )+ fv ( b ( ), 8) + fv( c ( ), 6)
s = sa + sb + sc   

PRINT s

END

FUNCTION fv (d(), n)

sd = 0

FOR i = 1 TO n

sd = sd + d(i)

NEXT i

fv = sd

END FUNCTION

SUB vv (d(), n)

FOR i = 1 TO n

INPUT d(i)

NEXT i

END SUB

Главным различием между процедурами SUB и FUNCTION  является то, что FUNCTION возвращает всегда только одиночное значение, тогда как SUB может вернуть в вызывающую программу несколько вычисленных значений. Это обстоятельство необходимо учитывать при использовании этих видов процедур.

4.3. Другие виды модулей

GOSUB-подпрограммы

Используются для организации подпрограмм без параметров.

Такая подпрограмма вызывается оператором:

GOSUB  метка

Описание  такой подпрограммы заключено между меткой и оператором возврата в вызывающую программу RETURN:

метка 

операторы  подпрограммы

RETURN
Например, приведенный фрагмент программы включает для  вывода горизонтальных разделительных линий для расчерчивания таблицы GOSUB-подпрограмму:

  . . .
 GOSUB m

 . . .
m: FOR i = 1  TO 70

PRINT “-”;

NEXT i

RETURN

Примечание. GOSUB-подпрограмма является устаревшей конструкцией и с успехом может быть заменена процедурой SUB.
Функции DEF FN
Функция DEF FN  может быть описана в линейной или блочной форме. 

Линейная форма записи функции DEF FN :

DEF FNимя [ ( параметры ) ]  =  выражение

Здесь  FNимя  - имя функции, начинающееся с символов FN

Блочная форма записи функции DEF FN :

 DEF FNимя  [ ( параметры ) ]

[ операторы ]

FNимя = выражение

END DEF

Например, следующий фрагмент программы использует функцию DEF FN (линейную форму ) для вычисления выражения:

. . .
y = ABS ( a ) + FNR ( x , z)

. . .

DEF FNR ( k , n ) = k ^ 2 - SQR ( k + n )

. . . 
Примечание. Функция DEF FN  не может находиться внутри другого  определения DEF FN и не может быть рекурсивной ( вызывать сама себя). Эта функция может вызываться только из того модуля, в котором она определена. Эти ограничения сняты в более современной конструкции  – процедуре FUNCTION.

5. ОБРАБОТКА ДАННЫХ РАЗНЫХ ТИПОВ В QBASIC

5.1. Типы данных, используемые в QBASIC

Переменные, используемые в QBASIC, могут быть числовыми и символьными. Для задания переменных различных типов к их именам добавляется специальный символ (суффикс). Числовые переменные бывают целыми, длинными целыми, обычной точности и двойной точности: 

· Целые (INTEGER) занимают в памяти 2 байта и используются для значений в диапазоне от – 32768 до + 32767 или от (–215) до 
(215 -1). Переменные этого типа называют короткими целыми. Для их задания используется суффикс «%». Например,   a% = 1

· Длинные целые (LONG) занимают в памяти 4 байта, используются для значений в диапазоне: от – 2147483648 до +2147483647 или от (–231) до (231 – 1). Используются переменные данного типа в тех случаях, когда требуется выполнить операции с целочисленными переменными, выходящими за рамки диапазона коротких целых чисел. Для задания длинных целых переменных используется суффикс «&». Например,  T& = 89365

· Обычной точности (SINGLE) требуют для своего размещения 4 байта. Используются для значений в диапазоне
 от –3.402823 E +38 до -1.40129 E – 45 для отрицательных значений и от +1.40129 E – 45 до +3.402823 E +38 для положительных значений.
Примечание. Если тип числовой переменной не задан, то она автоматически
(по умолчанию) становится переменной обычной точности. До сих пор мы этой возможностью все время пользовались и не уточняли тип числовых данных.

Задать переменную обычной точности можно, используя суффикс «!» или не используя суффикс вообще.

Например: 
        S! = 243. 67
PRINT S  ‘ будет выведено число 243. 67

· Двойной точности (DOUBLE) занимает в памяти 8 байт. Используется для значений в диапазоне 
от – 1.797693134862316 E +308  до -4.94965 E -324 для отрицательных значений и от  4.94965 E - 324 до 1.797693134862316 E +308 для положительных значений. Этот тип используется для точных математических вычислений, не допускающих потерю значности. Для задания переменной двойной точности используется 
суффикс «#». 
Например: D# = 53.49852436

Символьные переменные (строки) могут включать в себя любые символы из таблицы ASCII. Для присваивания значения символьной переменной значение заключается в двойные кавычки. Символьные переменные бывают переменной и фиксированной длины.

Примечание. В данном пособии строки фиксированной длины не рассматриваются.

· Строка переменной длины (STRING) – это последовательность длиной до 32567 символов из таблицы ASCII. В памяти она занимает столько байт, какова ее длина. Задать строку переменной длины можно, используя суффикс «$».

Например:  stroka$ = “программа”
5.2. Обработка целочисленных данных

Для решения группы задач, связанных с обработкой целых чисел, рассмотрим дополнительно следующие арифметические операции языка QBASIC:

Операция целочисленного деления: \
Например:   a = 10 : b = 4

                     y = a \ b
                     PRINT y ‘  Получим результат  2.

Операция нахождения остатка от целочисленного деления: MOD
Например:    x = 20 MOD 6

                      PRINT  x ‘  Результат  2.
                      PRINT 19  MOD  6.7 ‘ Результат  5.

Примечание. Приоритет выполнения операции целочисленного деления ниже приоритета операции  “*” и “/”, но выше приоритета операции MOD. Приоритет выполнения арифметических операций будет следующим:

1. возведение в степень (^);

2. умножение и деление (*,/);

3. целочисленное деление (\);

4. остаток после целочисленного деления (MOD);

5. сложение и вычитание (+,–).

При обработке целочисленных данных используются следующие встроенные функции округления:

INT(x)  возвращает первое целое, меньшее или равное числовому выражению  x.

Например:       a = INT(5.5)

                         PRINT a ’результат  5.

                  b = INT (–5.5)

                      PRINT b ‘результат – 6.  

FIX(x)  возвращает целую часть числового выражения x.

Например:        a = FIX (5.5)

                          PRINT a ‘результат  5.

                   b = FIX (–5.5)

                       PRINT b ‘ Результат –5.
Рассмотрим несколько задач на «целые числа».

	Задача 16.  Определить количество цифр введенного целого числа.
	Задача 17. Определить сумму цифр введенного целого числа.

	REM программа решения задачи 16
INPUT N%

y = ABS(N%): k = 0

WHILE y <> 0

k = k + 1

y = INT(y / 10)

WEND

PRINT "Количество цифр в числе="; k END


	REM программа решения задачи 17

INPUT N%

y = ABS(N%): s = 0

WHILE y <> 0

x = FIX(y / 10)

z = y - x * 10

s = s + z

y = x

WEND

PRINT "Сумма цифр = "; s

END


	Задача 18. Вывести в обратном порядке введенное целое число. Использовать подпрограмму.
	Задача 19. Определить наибольший общий делитель двух целых чисел. Решить задачу с использованием функции.

	DECLARE SUB naoborot (m%, l%)

REM программа решения задачи 18

INPUT N%

CALL naoborot(N%, k%)

PRINT k%

END

SUB naoborot (m%, l%)

l% = 0

y = ABS(m%)

WHILE y <> 0

z = y - INT(y / 10) * 10

l% = l% * 10 + z

y = INT(y / 10)

WEND

END SUB


	DECLARE FUNCTION naibdel% (a%, b%)

REM программа решения задачи 19

INPUT n%, m%

n%= ABS(n%): m%= ABS(m%)

k% = naibdel%(n%, m%)

PRINT "наиб. общий делитель=";k%

END

FUNCTION naibdel% (a%, b%)

IF a% < b% THEN

c% = a%

ELSE

c% = b%

END IF

FOR i% = 1 TO c%

IF a% MOD i% = 0 THEN

IF b% MOD i% = 0 THEN

d% = i%

END IF

END IF

NEXT i%

naibdel% = d%

END FUNCTION


5.3. Обработка символьных данных
Над строками можно осуществлять такие действия, как конкатенация и сравнение.

Конкатенация – это сложение символьных строк.

Например:

A$ = “q” : B$ = “basic”

PRINT  A$ + B$  ‘результат  “qbasic”
Сравнение строк производится при помощи операторов сравнения  =, <>, <, >, <=, >=  в соответствии с ASCII кодами символов в сравниваемых строках. Если ASCII коды равны, то строки равны. Строка, символы которой имеют меньшие ASCII коды или короче другой, считается меньшей. При сравнении учитываются начальные  и конечные пробелы.

Например:

“AA” < “BB”

“Qbasic” = “Q” + “basic”

“X&” > “X#”

“123” < “23^”

Примечание. При сравнении строк, содержащих русские буквы, не следует целиком полагаться на сравнение средствами Basic.

При обработке строк можно использовать следующие встроенные функции языка Qbasic.

INSTR ([n,]a$,b$) возвращает позицию первого вхождения строки b$ в строке a$. n устанавливает позицию начала поиска. Если это число не задано, то поиск начнется с первого символа строки. 
1<= n <= 32767.

Например:       y$=“задача”

                         x$=“да”

                         k = INSTR ( y$ , x$ )

                         PRINT   k ‘ будет выведено число  3.

LEN(a$)  возвращает количество символов строки a$.

Например:        a$=“задача”

                          y=LEN ( a$ )

                          PRINT y  ‘ будет выведено число 6.

LEFT$(a$,n$)  возвращает строку, содержащую n левых символов строки a$. 0<=n<=32767. Если n=0, то возвращается пустая строка.

Например:        y$ = “программа”

                          x$ = LEFT$( y$ , 3 )

                          PRINT  x$ ‘ будут выведены символы  “про”

RIGHT$(a$,n) возвращает n правых символов строки a$. 0<=n<=32767.

Например:       Y$ =“QBASIC”

                         X$=  RIGHT$(Y$,5)

                         PRINT X$ ‘ будут выведены символы “BASIC”

MID$(a$,n[,l]) - возвращает фрагмент строки a$ от символа в позиции n на заданную длину l (n и l в диапазоне от 1 до 32767). Если длина не указана, то возвращается фрагмент от начала до конца строки. Если n больше, чем длина символьного выражения, функция возвращает  пустую строку.

Например:        Y$=”ЗАДАЧА”

                           X$=MID$(Y$,3,2)

                           PRINT X$ ‘ будут выведены символы “ДА”.

Примечание. В QBASIC  существует оператор MID$, который заменяет фрагмент строки на содержание другой строки.

Формат оператора MID$ :

MID$(a$,n[,l]) = “символьное выражение”

символьное выражение  может быть переменной, константой, выражением.

Здесь a$  может быть только символьной переменной.

Если l отсутствует, то производится замена полным содержимым символьного выражения. Если l задана, то производится замена указанным количеством символов слева из символьного выражения. Заменяемый фрагмент не может быть больше текущей длины переменной.

LCASE$(a$)  возвращает символьную строку, в которой все латинские буквы преобразованы в строчные.

Например:  a$ = LCASE$( “Qbasic”)

                     PRINT  a$  ‘ Выводит  “qbasic”

UCASE$(a$)  возвращает символьную строку, в которой все латинские буквы – заглавные.

Например:  a$ = UCASE$( “Qbasic”)

                     PRINT  a$  ‘ Выводит  “QBASIC”

LTRIM$(a$) – символьная функция, возвращающая копию строки a$ с удаленными начальными пробелами.

RTRIM$(a$)  возвращает символьную строку с удаленными правыми пробелами.

SPACE$(n)  возвращает строку пробелов длиной n;  1<=n<=32767.

STRING$(m,n) – символьная функция, возвращает строку, заполненную m символами данного ASCII-кода.

0<=n<=255, ASCII-код символьного данного.

STRING$(m, символьное выражение)

символьное выражение - строка, первый символ которой используются для заполнения строки.

STR$(числовое выражение)  возвращает символьное представление числа.

ASC(a$)  возвращает числовое значение ASCII-кода первого символа в символьном выражении.

CHR$(код)  возвращает строку из одного символа, соответствующего ASCII-коду (код), 0<=код<=255, один из ASCII-кодов.

VAL(a$)  переводит строковое представление числа a$ в числовое значение обычной точности.

Задача 20. Определить количество вхождений заданного слова во введенном предложении.

REM программа решения задачи 20

INPUT "Введите предложение ", a$

INPUT "Введите слово ", b$

k = 1: n = 0

WHILE INSTR(k, a$, b$) > 0

m = INSTR(k, a$, b$)

n = n + 1

k = m + 1

WEND

PRINT "Количество искомых слов "; n

END

Задача 21. Подсчитать количество гласных букв в предложении.

REM программа решения задачи 21

INPUT "Введите предложение  ", a$

b$ = "аяэеоёуюыи"

k = 0

FOR i = 1 TO LEN(a$)

c$ = MID$(a$, i, 1)

IF INSTR(b$, c$) > 0 THEN

k = k + 1

END IF

NEXT i

PRINT "Количество гласных букв "; k

END

Задача 22. Вывести введенное предложение заглавными буквами. Напишем текст программы решения этой задачи с использованием подпрограммы:

DECLARE SUB zagl (x$, y$)

REM программа решения задачи 22

INPUT "Введите предложение : "; a$

CALL zagl(a$, b$)

PRINT b$

END

SUB zagl (x$, y$)

z$ = "ЁЙЦУКЕНГШЩЗХЪФЫВАПРОЛДЖЭЯЧСМИТЬБЮ"

t$ = "ёйцукенгшщзхъфывапролджэячсмитьбю"

y$ = ""

FOR i = 1 TO LEN(x$)

s$ = MID$(x$, i, 1)

IF INSTR(t$, s$) > 0 THEN

k = INSTR(t$, s$)

n$ = MID$(z$, k, 1)

ELSE

n$ = s$

END IF

y$ = y$ + n$

NEXT i

END SUB

Задача 23. Вывести слова введенного предложения в порядке убывания количества символов в них. Решим эту задачу с использованием функции.

DECLARE FUNCTION pf$ (x$)

REM программа решения задачи 23

CLS   ' очистка экрана

INPUT a$

y$ = pf$(a$)

PRINT y$

END

FUNCTION pf$ (x$)

REM определение количества слов в предложении

i = 1

WHILE INSTR(i, x$, " ") > 0

t = t + 1

i = INSTR(i, x$, " ") + 1

WEND

t = t + 1

REM формирование символьного массива, каждым элементом которого

REM является слово предложения

DIM c$(t)

k = 1

WHILE INSTR(k, x$, " ") > 0

y = INSTR(k, x$, " ")

m = y - k

b = b + 1

c$(b) = MID$(x$, k, m)

k = y + 1

WEND

IF y < LEN(x$) THEN

b = b + 1

k = LEN(x$) - y

c$(b) = RIGHT$(x$, k)

END IF

REM упорядочение элементов массива по убыванию количества букв в REM словах

DO

m = 0

FOR i = 1 TO t - 1

IF LEN(c$(i)) < LEN(c$(i + 1)) THEN

SWAP c$(i), c$(i + 1)

m = 1

END IF

NEXT i

LOOP UNTIL m = 0

REM формирование строки из упорядоченных слов

x$ = ""

FOR i = 1 TO t

x$ = x$ + c$(i) + " "

NEXT i

pf$ = RTRIM$(x$)

END FUNCTION

5.4. Построение сложных логических выражений

В ряде случаев требуется объединить несколько простых логических выражений (условий) в одно. Такое объединение выполняется на основе использования логических операций.

В QBASIC имеется 6 логических операций: NOT, AND, OR, XOR, EQV, IMP, но наиболее часто используются первые три.

Результатом выполнения логического выражения может быть ложное или истинное значение:  (ДА - истина) или (Нет - ложь).

Примечание. В QBASIC истина соответствует ненулевому значению (-1), а ложь - нулевому значению (0).

Действие логических операций показано в табл.5.

                                                                                                         Таблица 5

Действие логических операций.

	A
	B
	NOT A
	AND
	OR
	XOR
	EQV
	IMP

	И
	И
	Л
	И
	И
	Л
	И
	И

	И
	Л
	Л
	Л
	И
	И
	Л
	Л

	Л
	И
	И
	Л
	И
	И
	Л
	И

	Л
	Л
	И
	Л
	Л
	Л
	И
	И


Оператор NOT  (отрицание). NOT A истинно тогда, когда A 
ложно.

Оператор AND (логическое умножение). A AND B истинно тогда, когда истинно A и истинно B.

Оператор OR (логическое сложение). A OR B истинно тогда, когда хотя бы одно из A и B истинно.

Оператор XOR (исключающее или). A XOR B истинно тогда, когда  A и B имеют одинаковое значение.

Оператор EQV  (эквивалентность). A EQV B истинно тогда, когда A и B одновременно истинны или одновременно ложны.

Оператор IMP  (импликация). A IMP B ложно только тогда, когда A истинно, а B ложно. Во всех других случаях истинно.

Если в одном логическом выражении используется несколько логических операторов, то оно вычисляется в соответствии с приоритетами:

1) NOT, 2)AND, 3)OR, 4)XOR, 5)EQV, 6)IMP.

Для изменения установленного порядка вычислений используются круглые скобки.

Пример использования сложного логического выражения:

REM программа иллюстрирует использование сложного логического 

REM выражения и выводит на экран значение "Ложь"

x = 2: y = 11

IF  y / x = y \ x  OR  y MOD x <> 5  AND  NOT  INT(y / x) = 5  THEN

PRINT "Истина"

ELSE

PRINT "Ложь"

END IF

END

В результате вычисления логического выражения в соответствии с приоритетом операций оно принимает значение ложь:

ложь OR истина AND NOT истина
                                                   ложь

                                     ложь

                   ложь

6. ВВОД-ВЫВОД ДАННЫХ В QBASIC
В главе 2 мы познакомились с основными возможностями оператора ввода с клавиатуры INPUT и оператора вывода на экран PRINT. Теперь рассмотрим некоторые дополнительные средства ввода-вывода данных QBASIC.

6.1. Чтение из блока данных
В тех случаях, когда программе нужны фиксированные, заранее заданные значения, следует использовать операторы DATA, READ, RESTORE.

Оператор DATA
В этом операторе указываются значения для чтения последующими операто​рами READ.

Формат оператора:
                               DATA список констант

Операторы DATA задаются в любом месте программы (главного модуля) независимо от места задания оператора READ. Если в программе содержится несколько операторов DATA, то их можно рассматривать как объединенный оператор DATA, а все константы – объединенными в один список.

Примечание. Операторы DATA  могут быть заданы только на уровне главного модуля в любом его месте. Они не могут использоваться в процедурах.

Оператор READ

Считывает значения из оператора DATA  и присваивает их переменным, перечисленным в списке.

Формат оператора:

                              READ список переменных

Константы последовательно выбираются из списка констант оператора DATA и присваиваются переменным, указанным в операторе  READ. Тип константы в списке DATA должен соответствовать типу переменной в списке оператора READ, которой присваивается значение константы. Очередной оператор READ выбирает константы не с начала списка, а с того места в списке, на котором остановился предыдущий оператор READ. Неиспользованная константа из списка DATA игнорируется, а случай, когда в списке констант недостаточно для всех переменных  READ, является ошибкой.

Оператор RESTORE

Устанавливает точку отсчета для чтения данных из блоков DATA.

Формат оператора:

                              RESTORE [метки]

Позволяет READ заново считать значение в указанном операторе DATA. Здесь метка – это метка строки оператора DATA. Если метка строки не указана, сле​дующий оператор READ считывает значения из оператора DATA, расположенного в программе первым.

Например: 

                      READ a%, b%, d$, t

                      DATA 5, 10, ”slovo”, 0.25

                      READ c%, k%, l$, m$

                mt:DATA 15, 20, ”bukva”, ”znak”, 100, 200

                      RESTORE mt

                      READ n%(1), n%(2), p$(1,1), p$(1,2)

6.2. Форматированный вывод на экран

Форма представления данных, выводимых оператором PRINT, определяется значением самого данного. Числовые данные в зависимости от их значений преобразуются к десятичному числу (числу с фиксированной точкой) или к числу, представленному в экспоненциальной форме. Длина поля, занимаемого данным, также определяется его значением. Каждый символ строкового данного представляется одной позицией на экране.

Для управления формой представления выводимых данных используется оператор PRINT USING.

Оператор PRINT USING
При использовании оператора PRINT USING форма представления данного определяется шаблоном (форматом), задаваемым в этом операторе, и не зависит от значения данного. 

Форма записи этого оператора:

                      PRINT USING формат;список выражений [{,l;}]

формат – символьное выражение, содержит специальные символы, управляю​щие представлением выводимых данных;

список выражений  – числовые или строковые выражения, для разделения которых используется запятая.

Для вывода числовых значений используются следующие форматы:

# – цифровая позиция. Если выводимое число имеет меньше цифр, чем указано шаблоном, то оно выравнивается вправо;

. – десятичная точка. Слева располагается целая часть, а справа – дробная. Лишние дробные знаки отсекаются с округлением;

+ – задает позицию знака числа (+ или –), либо перед числовым полем, либо после. Может быть либо первым, либо последним символом шаблона.

^^^^  или ^^^^^ определяют экспоненциальную форму представления числа и могут быть указаны после цифровых позиций.

Примечание. Если выводимое число не помещается в поле формы, то перед  ним выводится знак %.

Примеры:

                     PRINT USING “##.##”;0.25                (0.25)

                     PRINT USING “###.##”;126.783         (126.78)

                     PRINT USING “##.##”;11.3,6.8           (11.30 6.80)

                     PRINT USING “+##.##”;-62.84 4.5      (–62.84 +4.5)

                     PRINT USING “##.##^^^^”;234.56     (2.35E +02)

                     PRINT USING “##.##^^^^^”;234.56   (2.35E +002)

                     PRINT USING “##.##”;111.22             (%111.22)

Рассмотрим фрагмент программы вывода  матрицы (в n строк по m чисел в каждую), содержащей  целочисленные значения:

. . .

DIM  a%( n, m )

. . . 

FOR i = 1 TO n

PRINT

FOR j = 1 TO m

PRINT USING  “#####” ; a% ( i , j ) ;

NEXT j

NEXT i

. . .

Для вывода символьных выражений используются форматы:

! –    вывод первого символа указанной строки;

\ \ –  вывод 2+n символов из строки, где n – число пробелов между символами (\). Если поле вывода больше длины строки, то строка выравнивается влево, а справа дополняется пробелами;

&  – вывод строки без модификаций.

Примеры:

                     PRINT USING “!”;Киев                 (К)

                     PRINT USING “\ \”;Париж             (Пар)

                     PRINT USING “&”;Волгоград      (Волгоград)

6.3. Очистка экрана, управление курсором и установка
 экранных цветов

Оператор  CLS

В результате выполнения работы этого оператора информация, находящаяся на экране, стирается, курсор размещается в левом верхнем углу экрана.

Оператор  LOCATE

Оператор LOCATE обеспечивает возможность вывода данных, начиная с любой позиции экрана.

Оператор имеет следующий формат:

LOCATE [строка] [, [столбец]]

строка – номер строки на экране, в которую будет перемещен курсор (число 1–25);

столбец – номер столбца на экране, в который будет перемещен курсор (число 1–80);

Если опустить аргументы строка и столбец, курсор остается в позиции, установленной предыдущим оператором LOCATE или предыдущим оператором ввода-вывода.

Например, оператор LOCATE 10, 20 устанавливает курсор в 20 колонку 10 строки.

Оператор COLOR

Оператор, устанавливающий экранные цвета, имеет следующий формат:

COLOR [основной ] [, [ фоновый  ] [окаймление] ]

основной – цвет текста (0 – 31, где 16 – 31 – мерцающие);

фоновый – цвет экрана (0 – 7);

окаймление – цвет вокруг  экрана (0 – 15).

Номера некоторых экранных цветов:

	0 –
черный
	5 –
розовый
	10 –
светло-зеленый

	1 –
синий
	  6 –
коричневый
	11 –
светло-голубой

	2 –
зеленый
	  7 –
белый
	12 –
светло-красный

	3 –
голубой
	8 –
серый
	13 –
светло-розовый

	4 –
красный
	9 –
светло- синий
	14 –
желтый 15 –  ярко -белый


Действие оператора распространяется на символы, которые будут выведены в дальнейшем, при этом цвет символов, уже находящихся на экране, не изменяется. В текстовом режиме на экране одновременно могут находиться символы различного цвета.

Например, оператор COLOR 7, 3  устанавливает белый текст на голубом фоне, т.е. символы, выводимые последующими операторами, изображаются белым цветом на голубом фоне.

6.4. Создание и обработка последовательных файлов

Файлом называется определенный набор данных, который рассматривается как одно целое и либо заносится на МД, либо считывается с него.

В зависимости от способа организации данных различают файлы двух типов: с последовательным и произвольным доступом.

В файлах с последовательным доступом данные размещаются в порядке их поступления в файл. При чтении такого файла данные становятся доступными в том порядке, в котором они были записаны. В этом случае при поиске нужных данных файл нужно просматривать от начала до конца.

В файле с произвольным доступом данные организуются таким образом, что доступ к ним может быть осуществлен в любом порядке независимо от того, в какой последовательности данные помещаются в файл.

При работе с файлами различают две основные и две вспомогательные операции. К основным относятся ввод и вывод (INPUT# и PRINT# или WRITE#). К вспомогательным – открытие файла на диске и закрытие его (OPEN и CLOSE).

Оператор OPEN
 С этого оператора начинается обработка любого файла. Оператор OPEN открывает файл для ввода (или вывода).

Оператор открытия файла выполняет следующие функции:

· связывает с файлом номер, который используется в операторах ввода-вывода для задания обрабатываемого файла;

· определяет организацию данных в файле и режимы его обработки.

Выводные файлы на дисках могут обрабатываться как создаваемые вновь и расширяемые.

Формат оператора:

                        OPEN файл [FOR режим] AS[#] номер

режим  это один из следующих режимов:

INPUT – (последовательный ввод из файла), обрабатываемый;

OUTPUT – (последовательный вывод в файл), создаваемый;

APPEND – (последовательный вывод с добавлением, т. е. установка указателя записи к концу файла), расширяемый.

номер – целочисленное выражение от 1 до 255, номер файла (логический номер файла). Заданный номер должен использоваться при всех последующих обращениях к файлу.

файл – задается строковым выражением и определяет имя файла и устройство, на котором он расположен.

Оператор CLOSE

После завершения обработки файла необходимо осуществить стандартные действия, связанные с освобождением ресурсов, используемых файлом. 

Выполнение этих действий называется закрытием файла и осуществляется оператором CLOSE.

Формат оператора:

CLOSE [[#]номер-файла[,[#]номер-файла]....]

номер-файла – логический номер открытого файла. Если все аргументы опущены, закрываются все файлы и устройства. После закрытия файл может быть повторно открыт для последующей обработки, а номер файла может использоваться при открытии других файлов.

Оператор INPUT#

 Читает данные из файла или последовательного устройства и присваивает их переменным.

Формат оператора:

                        INPUT# номер-файла, переменные

номер-файла – логический номер файла;

переменные – список переменных, которым присваиваются значения, читаемые из файла. Типы данных должны соответствовать друг другу. 

Оператор PRINT#

 Записывает данные в последовательный файл.

Формат оператора:

                        PRINT# номер-файла, список выражений[,|;]

номер-файла – логический номер файла, открытого оператором OPEN;

список выражений – содержит значения, записываемые в файл. Если он пуст, то записывается пустая строка. Пробелы между элементами списка записы​ваются в файл.

Оператор WRITE#

Записывает данные в последовательный файл.

Формат оператора:

WRITE # номер файла [ , список - выражений]

номер-файла – логический номер файла, открытого оператором OPEN;

список выражений – одно или несколько выражений, разделенных запятыми, значения которых выводятся в файл. Выводимые значения разделяются запятыми, символьные строки заключаются в кавычки.

Функция EOF (номер файла) 

Определяет условие конца файла. Возвращает  –1(“истина”), если указатель в файле последовательного доступа дошел до его конца. Используется для тестирования конца файла при вводе данных из файла.

Задача 24.  Создать последовательный файл на диске с именем “OZEN.DAT”, содержащий информацию о сдаче студентами экзамена. Количество обрабатываемой информации может быть произвольным.

REM программа решения задачи 24

CLS

OPEN "OZEN.DAT" FOR OUTPUT AS #1

INPUT "Фамилия ", fio$

INPUT "Оценка ", oz

WHILE fio$ <> ""

PRINT #1, fio$; “,”;oz

INPUT "Фамилия ", fio$

INPUT "Оценка ", oz

WEND

CLOSE #1

END

Задача 25. Обработать информацию о сдаче студентами экзамена, содержащуюся в последовательном файле на диске с именем «OZEN.DAT», с целью вычисления средней оценки.

REM программа решения задачи 25

CLS

OPEN "OZEN.DAT" FOR INPUT AS #1

WHILE NOT EOF(1)

INPUT #1, fio$, oz

s = s + oz

k = k + 1

PRINT fio$, oz

WEND

CLOSE #1

IF k <> 0 THEN

PRINT s / k

END IF

END

ПРИЛОЖЕНИЕ 1. 

Инструкция по работе с QBASIC

1. Загрузить QBASIC. Для этого активизировать каталог C:\DOS и загрузить файл qbasic.exe. Появится окно с предложением воспользоваться справочной системой.

2. Для отказа от работы со справочной системой нажать клавишу ESC, после чего появляется окно встроенного редактора QBASIC.

3. Набрать текст программы. После набора каждой строки нажимать клавишу ENTER.

4. Выполнить набранную программу. Для этого:

· активизировать меню нажатием клавиши ALT;

· установить курсор на позицию меню RUN и нажать клавишу ENTER;

· установить курсор на пункт START и нажать клавишу ENTER;

· в ответ на появившийся "?" набрать требуемое количество вводимых данных (при вводе нескольких данных они должны разделяться запятыми). Количество вводимых данных определяется количеством элементов в списке ввода оператора INPUT;

· ознакомиться с результатами решения задачи и нажать любую клавишу для возврата в экран QBASIC.

5.  Набрать текст следующей программы. Для этого:

· активизировать меню нажатием клавиши ALT;

· установить курсор на позицию меню FILE и нажать клавишу ENTER;

· установить курсор на пункт меню NEW и нажать клавишу ENTER; 

· в ответ на запрос о сохранении на диске предыдущей программы нажать клавишу "N";

· повторить все действия, указанные в пунктах 3,4.

6. Выйти из программы QBASIC. Для этого:

· активизировать меню нажатием клавиши ALT;

· установить курсор на позицию меню FILE и нажать клавишу ENTER;

· установить курсор на пункт меню EXIT и нажать клавишу ENTER.

Некоторые команды редактирования:

Удаление символа в позиции курсора - клавиша  DEL

Удаление символа слева от курсора - клавиша BACKSPACE

Удаление строки текста - одновременно клавиши CTRL и Y

Переключение режимов вставка / замена - клавиша INSERT

